### 3.3 *Intégration*

Pour intégrer une expression, on utilise la fonction **int**. Cette même fonction va également permettre de trouver une primitive de cette expression en n'indiquant pas de bornes d'intégration. Il y a donc deux syntaxes pour la fonction **int** :

> Pour trouver une primitive d'une expression :
> **>int(*expression*,*var*);**
>
> Pour intégrer une expression entre $a$ et $b$ :
> **>int(*expression*,*var*=*a*..*b*);**

Par exemple :

```
>int(2*x^2+3*cos(x)+ln(x/(x+1)),x);
```

$$\frac{2}{3}x^3+3\sin(x)+\ln\left(-\frac{1}{x+1}\right)+\ln\left(1-\frac{1}{x+1}\right)\left(1-\frac{1}{x+1}\right)(x+1)$$

On peut remarquer que Maple ne rajoute pas de constsnte d'intégration afin de simplifier l'expression finale.

On peut aussi, comme on l'a dit, intégrer entre deux bornes $a$ et $b$ :

```
>int(cos(x)+sin(ln(x)),x=0..10);
```

$$\sin(10)+5\sin(\ln(2)+\ln(5))-5\cos(\ln(2)+\ln(5))$$

On obtient alors sa valeur approchée à l'aide de la fonction **evalf** :

```
>evalf(");
```

$$6.516888125$$

Maple sait aussi gérer les intégrales généralisées :

```
>int((sin(x)/x)^2,x=0..infinity);
```

$$\frac{1}{2}\pi$$

On peut remarquer que, lorsque Maple ne réussit pas à calculer une intégrale, il renvoie la forme inerte :

```
>int(1/(1-x),x=0..2);
```

$$\int_0^2 \frac{1}{1-x}\,\mathrm{d}x$$

Signalons également que Maple permet d'afficher la forme inerte de l'intégrale si l'on entre la fonction **Int** (i majuscule). Par exemple :

```
>Int(ln(x)/x,x);
```

$$\int \frac{\ln(x)}{x}\,\mathrm{d}x$$

On peut alors très simplement obtenir la valeur de cette intégrale en utilisant la fonction **value** :

```
>value(");
```

$$\frac{1}{2}\ln(x)^2$$

Le package **student** offre entre autres deux fonctions (**intparts** et **changevar**) qui permettent de changer de variable et d'intégrer par parties. Par exemple :

```
>J:=Int(ln(x),x);
```

$$J := \int \ln(x)\,\mathrm{d}x$$

On effectue un changement de variable avec la fonction **changevar** du package **student** :

**>changevar(*var=expression,intégrale,var*);**

On commence donc par charger le package **student** :

```
>with(student):
```

Dans notre cas, cela donne :

```
>changevar(u=ln(x),J,u);
```

$$\int u\mathrm{e}^u\,\mathrm{d}u$$

On demande alors à Maple la valeur de cette primitive:

```
>value(");
```

$$u\mathrm{e}^u - \mathrm{e}^u$$

Ensuite, on revient à nos variables d'origine à l'aide de la fonction **subs** :

```
>subs(u=ln(x),");
```

$$\ln(x)\,e^{\ln(x)} - e^{\ln(x)}$$

et on simplifie[1] le résultat à l'aide de la fonction **simplify** :

```
>simplify(");
```

$$\ln(x)x - x$$

> On peut réaliser une intégration par parties à l'aide de la fonction **intparts** du package **student** :
>
> **>intparts(*intégrale,expression à dériver*);**
>
> On précise l'expression à dériver, c'est-à-dire $u$ dans la formule de l'intégration par parties : $\int uv' = [uv] - \int u'v$.

Notre intégrale $J$ peut être calculée à l'aide d'une intégration par parties :

```
>intparts(J,ln(x));
```

$$\ln(x)x - \int 1\,dx$$

On demande alors la valeur finale à Maple :

```
>value(");
```

$$\ln(x)x - x$$

## 3.4 *Développements limités et asymptotiques*

> La fonction **taylor** permet de réaliser des développements limités :
>
> **>taylor(*expression*,*var*=*point*,*ordre*);**

Par défaut, l'ordre est fixé à 6. On peut aussi noter que l'ordre du développement est le degré du premier terme qui est négligeable. Enfin, signalons que Maple emploie des $O(x)$ au lieu des $o(x)$ traditionnels.

Par exemple :

```
>taylor(sin(x),x=0,8);
```

$$x - \frac{1}{6}x^3 + \frac{1}{120}x^5 - \frac{1}{5\,040}x^7 + O(x^8)$$

Pour réutiliser le développement limité, il sera généralement nécessaire de le convertir en polynôme, c'est ce que l'on fait ci-dessous :

```
>convert(taylor(sin(x),x=0),polynom);
```

$$x-\frac{1}{6}x^3+\frac{1}{120}x^5$$

On pourra alors obtenir la courbe représentative de cette expression ou encore calculer sa valeur en différents points...

On peut également réaliser un développement limité en un point différent de 0 :

```
>taylor(sin(x),x=Pi,4);
```

$$-(x-\pi)+\frac{1}{6}(x-\pi)^3+O((x-\pi)^4)$$

On peut même obtenir la formule de Taylor-Young :

```
>taylor(f(x),x=a,4);
```

$$f(a)+D(f)(a)(x-a)+\frac{1}{2}(D^{(2)})(f)(a)(x-a)^2+\frac{1}{6}(D^{(3)})(f)(a)(x-a)^3+O((x-a)^4)$$

En ce qui concerne les développements asymptotiques, on utilise la fonction **asympt** :

**>asympt(*expression*,*var*);**

Par exemple :

```
>asympt(x/(1-x-x^2),x);
```

$$-\frac{1}{x}+\frac{1}{x^2}-\frac{2}{x^3}+\frac{3}{x^4}-\frac{5}{x^5}+O\left(\frac{1}{x^6}\right)$$

## 3.5 *Fonctions spéciales*

La fonction **piecewise** permet de générer une fonction par morceaux. Par exemple :

```
>f:=x->piecewise(x<-2,-1,x<2,1,3);
```

$$f:=x\rightarrow \text{piecewise}(x<-2,-1,x<2,1,3)$$

On peut représenter[1] cette fonction pour bien voir à quoi elle correspond :

```
>plot(f);
```




Il existe aussi la fonction **Heaviside** définie par : $Heaviside(x) = \begin{cases} 1 \text{ si } x \geq 0 \\ 0 \text{ si } x < 0 \end{cases}$

# 4. Expressions symboliques

Maple manipule les mêmes expressions que vous sur le papier. Cependant, il ne faut omettre ici aucun signe, comme ceux marquant la multiplication que l'on ne fait généralement pas figurer sur la feuille. Nous allons donc nous intéresser dans ce paragraphe aux différents types d'expressions symboliques.

## 4.1 *Polynômes*

On peut définir un polynôme :

```
>A:=x^2+3*x+1;
```

$$A := x^2 + 3x + 1$$

```
>B:=x-1;
```

$$B := x - 1$$

On peut tester si $A$ divise $B$ par l'intermédiaire de la fonction **divide** :

```
>divide(A,B,x);
```

$$false$$

On peut alors écrire ce polynôme sous la forme $A = B \cdot Q + R$. Commençons par calculer le quotient :

```
>quo(A,B,x);
```

$$x + 4$$

Calculons alors le reste :

```
>rem(A,B,x);
```

$$5$$

Bien sûr, on peut aussi développer et factoriser les polynômes :

```
>C:=(x-3)*(x-2);
```

$$C := (x-3)\,(x-2)$$

Par défaut, Maple ne développe donc pas les expressions.

> On développe une expression à l'aide de la fonction **`expand`** :
>
> **`>expand(`*`expression`*`);`**

Par exemple, pour le polynôme C définit précédemment :

```
>expand(C);
```

$$x^2 - 5x + 6$$

On peut alors retrouver l'expression d'origine de C en factorisant l'expression obtenue à la ligne précédente.

> On factorise une expression à l'aide de la fonction **`factor`** :
>
> **`>factor(`*`expression`*`);`**

Ainsi :

```
>factor(");
```

$$(x-3)\,(x-2)$$

Dans certains cas, on peut aussi vouloir factoriser sur $\mathbb{C}$ :

```
>E:=x^2+2*x+2;
```

$$E := x^2 + 2x + 2$$

```
>factor(E);
```

$$x^2 + 2x + 2$$

Par défaut, Maple factorise donc dans $\mathbb{R}$. On peut cependant forcer Maple à factoriser dans $\mathbb{C}$ en ajoutant l'argument **`complex`** dans la fonction **`factor`** :

```
>factor(E,complex);
```

$$(x + 1 + 1\,.\,I)\,(x + 1 - 1\,.\,I)$$

On peut entrer les polynômes de manière désordonnée :

```
>F:=x+3*x^2+x+x^4+x^3;
```

$$F := 2x + 3x^2 + x^4 + x^3$$

## 4.2 *Fractions rationnelles*

Une fraction rationnelle est de la forme : $\dfrac{a_n x^n + a_{n-1}x^{n-1} + \ldots + a_1 x^1 + a_0}{b_m x^m + b_{m-1}x^{m-1} + \ldots + b_1 x^1 + b_0}$. On peut donc commencer par extraire de cette fraction le numérateur et le dénominateur.

```
>F:=(3*x^4+4*x^2+2*x+1)/(x^3+2*x+1);
```

$$F := \frac{3x^4 + 4x^2 + 2x + 1}{x^3 + 2x + 1}$$

On extrait le dénominateur et le numérateur d'une fraction (rationnelle ou non) avec les fonctions **denom** et **numer** :

```
>denom(fraction);
>numer(fraction);
```

Par exemple sur la fraction $F$ que l'on vient de définir :

```
>denom(F);
```

$$x^3 + 2x + 1$$

```
>numer(F);
```

$$3x^4 + 4x^3 + 2x + 1$$

La fonction **expand** vue pour les polynômes produit une réponse peu intéressante :

```
>expand(F);
```

$$3\frac{x^4}{x^3 + 2x + 1} + 4\frac{x^2}{x^3 + 2x + 1} + 2\frac{x}{x^3 + 2x + 1} + \frac{1}{x^3 + 2x + 1}$$

La fonction **factor** en revanche factorise le numérateur et le dénominateur :

```
>G:=(x^2-x-6)/(x^2+1);
```

$$G := \frac{x^2 - x - 6}{x^2 + 1}$$

```
>factor(G);
```

$$\frac{(x+2)(x-3)}{x^2 + 1}$$

## L'essentiel

Pour évaluer numériquement une expression, on utilise la fonction **evalf** :
`>evalf(expression, nombre de chiffres);`

La fonction **floor** permet de prendre la partie entière d'un nombre :
`>floor(nombre);`

**iquo** et **irem** renvoient respectivement le quotient et le reste de la division euclidienne de $a$ par $b$ :
`>iquo(a,b);`
`>irem(a,b);`

La fonction **isprime** détermine si un nombre est ou n'est pas premier :
`>isprime(nombre);`

Pour générer un entier au hasard entre $a$ et $b$, on utilise la fonction **rand** :
`>nom:=rand(a..b);`

Ensuite, on utilise la procédure générée en tapant `nom();`

Les coefficients binomiaux sont donnés par **binomial** :
`>binomial(n,k);`

On somme sur les entiers à l'aide de la fonction **sum** :
`>sum(expression(var),var=a..b);`

**I** est le $i$ complexe.

On définit une fonction par :
`>fonction:=var->expression(var);`

On calcule une limite à l'aide de la fonction **limit** :
`>limit(expression(var),var=point);`

On dérive une expression à l'aide de la fonction **diff** :
`>diff(expression,var);`

On trouve une primitive d'une expression à l'aide de la fonction **int** :
`>int(expression(var),var);`

## L'essentiel (suite)

On calcule l'intégrale d'une expression entre $a$ et $b$ par :
`>int(expression(var),var=a..b);`

On effectue un développement limité avec la fonction **taylor** :
`>taylor(expression(var),var=point,ordre);`

Pour développer une expression, on utilise la fonction **expand** :
`>expand(expression);`

Pour factoriser une expression, on utilise la fonction **factor** :
`>factor(expression);`

Algèbre linéaire

## 1.1 *Définition d'une matrice*

La définition d'une matrice se fait à l'aide de la fonction **matrix** :

**>*nom*:=matrix(*n*,*m*);**

Crée une matrice vide à $n$ lignes et $m$ colonnes.

**>*nom*:=matrix([[$v_{1,1}$, $v_{1,2}$,..., $v_{1,m}$],..., [$v_{n,1}$,..., $v_{n,m}$]]);**

Crée une matrice qui contient les valeurs $v_{1,1}$,..., $v_{n,m}$.

On peut aussi définir cette même matrice par :

**>*nom*:=matrix(*n*,*m*, [$v_{1,1}$, $v_{1,2}$,..., $v_{1,m}$],..., [$v_{n,1}$,..., $v_{n,m}$]);**

## 1.2 *Opérations courantes sur les matrices*

Après avoir défini des matrices, nous allons voir comment les manipuler. On récapitule dans le tableau suivant les principaux opérateurs du calcul matriciel.

**Tableau 4 : Opérateurs du calcul matriciel**

| Opérateur | Notation |
|---|---|
| Somme | + |
| Différence | - |
| Produit | &* |
| Puissance | ^ ou ** |

Prenons un exemple de chaque opération. Pour cela, on commence par définir deux matrices :

```
>A:=matrix([[2,3,4],[6,1,0],[8,9,7]]);
```

$$A := \begin{bmatrix} 2 & 3 & 4 \\ 6 & 1 & 0 \\ 8 & 9 & 7 \end{bmatrix}$$

```
>B:=matrix([[5,2,0],[4,9,3],[8,7,1]]);
```

$$B := \begin{bmatrix} 5 & 2 & 0 \\ 4 & 9 & 3 \\ 8 & 7 & 1 \end{bmatrix}$$

On peut au passage vérifier si Maple reconnaît $A$ comme une matrice :

```
>whattype(A);
```

*string*

Maple reconnaît $A$ comme une chaîne de caractères... En fait, il faut évaluer la matrice pour que Maple reconnaisse $A$ comme telle :

```
>whattype(evalm(A));
```

*array*

Ainsi, pour sommer deux matrices :

```
>C:=A+B;
```

$$C := A + B$$

Apparemment, rien ne se produit…

Cependant, la somme a bien été réalisée, comme on peut le constater en évaluant C :

```
>evalm(C);
```

$$\begin{bmatrix} 7 & 5 & 4 \\ 10 & 10 & 3 \\ 16 & 16 & 8 \end{bmatrix}$$

Maintenant, intéressons-nous au produit matriciel. Il faudra bien se rappeler que l'opérateur Maple du produit matriciel n'est pas `*` mais `&*`. On a donc :

```
>E:=A&*B;
```

$$E := A \&^* B$$

Comme pour la somme, a priori, rien ne se passe. On peut cependant vérifier que $E$ est bien une matrice :

```
>evalm(E);
```

$$\begin{bmatrix} 54 & 59 & 13 \\ 34 & 21 & 3 \\ 132 & 146 & 34 \end{bmatrix}$$

Nous pouvons également élever une matrice à une puissance :

```
>F:=A^3;
```

$$F := A^3$$

```
>evalm(F);
```

$$\begin{bmatrix} 666 & 531 & 468 \\ 342 & 289 & 240 \\ 1\,476 & 1\,203 & 1\,071 \end{bmatrix}$$

Pour finir, on peut aussi appliquer une fonction à toute une matrice:

```
>sin(A);
```

$$\sin(A)$$

```
>evalm(");
```

$$\begin{bmatrix} \sin(2) & \sin(3) & \sin(4) \\ \sin(6) & \sin(1) & 0 \\ \sin(8) & \sin(9) & \sin(7) \end{bmatrix}$$